
確率密度分布―PROD 
 
 
 プログラム PROD（Probability Density Distribution）は，与えられたデータを，まずその最大値
で規準化したうえで，最大値と最小値の間を，級間 0.1，21 階数のそれぞれに属する標本値の数
をカウントし，確率密度分布を計算するブルーチン副プログラムである．データの平均値及び標
準偏差も同時に算出する． 
 
 

PROD（確率密度分布） 
 
【目 的】 
 与えられたデータの確率密度分布を計算し，データの平均値と標準偏差も同時に求める． 

 
 
【使用法】 
（1） 接続方法 
 CALL PROD (N，X，ND，RAMP，PROP，AV,  SD) 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
（2） 必要なサブルーチン及び関数副プログラム 

ない 
 
 
 

引　数 型 プログラムを呼ぶときの内容 プログラムから戻ったときの内容

N I  データの総数  不変

X
R

1次元配列

（ND）
 データ  最大値で規準化されたデータ

ND I  主プログラムにおけるXのディメンション  不変

RAMP
R

1次元配列

(21)
 何も入れなくてよい  相対振幅の級代表値

PROP
R

1次元配列

(21)
 何も入れなくてよい  各階級の確率分布(単位%）

AV R  何も入れなくてよい  最大値で規準化された平均値

SD R  何も入れなくてよい  最大値で規準化された標準偏差



PROD 
 

-2- 
 

【プログラム・リスト】 
 
C  * * * * * * * * * * * * * * * * * * * * * * * * * * *                PROD   1 

C     SUBROUTINE FOR PROBABILITY DENSITY DISTRIBUTION                   PROD   2 

C  * * * * * * * * * * * * * * * * * * * * * * * * * * *                PROD   3 

C                                                                       PROD   4 

C                                   CODED BY Y.OHSAKI                   PROD   5 

C                                                                       PROD   6 

      SUBROUTINE PROD(N,X,ND,RAMP,PROP,AV,SD)                           PROD   7 

C                                                                       PROD   8 

      DIMENSION X(ND),RAMP(21),PROP(21)                                 PROD   9 

      DIMENSION BOUND(21)                                               PROD  10 

C                                                                       PROD  11 

C     INTIALIZATION                                                     PROD  12 

C                                                                       PROD  13 

      DO 110 I=1,21                                                     PROD  14 

      RAMP(I)=1.-REAL(I-1)/10.                                          PROD  15 

      BOUND(I)=RAMP(I)-0.05                                             PROD  16 

      PROP(I)=0.                                                        PROD  17 

  110 CONTINUE                                                          PROD  18 

      XMAX=0.                                                           PROD  19 

      DO 120 M=1,N                                                      PROD  20 

      XMAX=AMAX1(XMAX,ABS(X(M)))                                        PROD  21 

  120 CONTINUE                                                          PROD  22 

      DO 130 M=1,N                                                      PROD  23 

      X(M)=X(M)/XMAX                                                    PROD  24 

  130 CONTINUE                                                          PROD  25 

      AV=0.                                                             PROD  26 

      SD=0.                                                             PROD  27 

C                                                                       PROD  28 

C     PROBABILITY DENSITY                                               PROD  29 

C                                                                       PROD  30 

      DO 150 M=1,N                                                      PROD  31 

      AV=AV+X(M)                                                        PROD  32 

      SD=SD+X(M)**2                                                     PROD  33 

      DO 140 I=1,21                                                     PROD  34 

      IF(X(M).LT.BOUND(I)) GO TO 140                                    PROD  35 

      PROP(I)=PROP(I)+1.                                                PROD  36 

      GO TO 150                                                         PROD  37 

  140 CONTINUE                                                          PROD  38 

  150 CONTINUE                                                          PROD  39 

      DO 160 I=1,21                                                     PROD  40 

      PROP(I)=PROP(I)/REAL(N)*100.                                      PROD  41 

  160 CONTINUE                                                          PROD  42 

      AV=AV/REAL(N)                                                     PROD  43 

      SD=SQRT(SD/REAL(N)-AV**2)                                         PROD  44 

      RETURN                                                            PROD  45 

      END                                                               PROD  46 

【使用例】 
 エル・セントロ地震波の確率密度分布を求める.主プログラムは，例えば次のようになり，アウ
ト・プットは表 3‐1に示したものが得られる． 
 
      CHARACTER NAME*50 

      DIMENSION X(800),RAMP(21),PROP(21) 

C 

      READ(5,501) NAME,DT,NN,(X(M),M=1,NN) 

      CALL PROD(NN,X,800,RAMP,PROP,AV,SD) 

      WRITE(6,601) NAME,(RAMP(I),PROP(I),I=1,21) 

      WRITE(6,602) AV,SD 

      STOP 

C 

 501  FORMAT(A50,F10.0,I10/(8F10.0)) 

 601  FORMAT(A50//T3,'-- PROBABILLITY DENSITY DISTRIBUTION --'// 

     *       T5,'REALT.AMP   PROB.DENSITY(PERCENT)'//(F13.2F16.5)) 

 602  FORMAT(/T5,'NORMALIZED AVERAGE VALUE',F10.3/ 

     *        T5,'NORMALIZED STANDARD DEV.',F10.3) 

      END 

 

アウトプット： 
 

EQ.01 - EL CENTRO, CALIF. 1940.5.18 NS             

 

  -- PROBABILLITY DENSITY DISTRIBUTION -- 

 

    REALT.AMP   PROB.DENSITY(PERCENT) 

 

         1.00            0.50 

         0.90            0.75 

         0.80            0.62 

         0.70            0.62 

         0.60            2.38 

         0.50            2.62 

         0.40            3.12 

         0.30            5.12 

         0.20            8.25 

         0.10           16.38 

         0.00           24.62 

        -0.10           13.00 

        -0.20            4.50 

        -0.30            3.62 

        -0.40            4.00 

        -0.50            4.12 

        -0.60            2.88 

        -0.70            1.25 

        -0.80            0.75 

        -0.90            0.88 

        -1.00            0.00 

 

    NORMALIZED AVERAGE VALUE    -0.000 

    NORMALIZED STANDARD DEV.     0.314 


